Cvicné ulohy pro jazyk Tiny BASIC (ver 1)
Priklad 1:

Zadani:
Vytvoite program pro vypocet plochy obdélnika. Po spusténi programu bude uZivatel vyzvan, at
zada délky stran. Pak se vypiSe velikost plochy. VSe pocitejte s celoCiselnymi hodnotami.

Rada:
Pouzijte ptikazy PRINT a INPUT. Plocha obdélnika se vypocita jako soucin délek jeho stran.

Priklad reSeni:

0 REM Vypocet plochy obdelnika
1 PRINT “Zadej delku stran:\n“

2 INPUTA; INPUT B

3 PRINT “Plocha je: “, A*B

Poznamky k FeSeni:

Priklad by se dal napsat i na jeden Fadek:
1 INPUT A; INPUT B; PRINT “Plocha je: “, A*B

Priklad 2:

Zadani:
Vytvorte program pro vypocet plochy kruhu s pfesnosti na 2 desetinna mista. Po spusSténi programu
bude uZivatel vyzvan, at’ zada polomér kruhu a pak se vypiSe velikost plochy.

Rada:

PouZijte prikazy PRINT, LET, IF-THEN a INPUT. Tiny BASIC pocita pouze s celymi Cisly.
PouZijte pro vypocet plochy hodnotu 1 jako 314 a pak zobrazte vysledek postupné. Nejprve vypisSte
celou hodnotu ziskanou délenim 100, znak pro desetinnou ¢arku a pak zbytek po déleni 100.
Nezapomeiite, Ze zbytek po déleni je nutné zobrazit i s prvni nulou — napr. 08.

Vzorec pro vypocet plochy je znamy:

S = 7

Priklad feSeni:

0 REM Vypocet plochy kruhu

1 PRINT “Zadej polomer:\n*

2 INPUTR

3 LET S=R*R*314; LET C=S/100
4 PRINT “Plocha je: “, C, “.“

5 LET Z=S-C*100

6 IF Z<10 THEN PRINT “0*

7 PRINT Z

Poznamky k reSeni:

Tiny BASIC nema operaci pro mocniny cisel. Proto je druha mocnina zapsana jako R*R (fadek 3).
A také nepracuje s desetinnymi Cisly. Nasobenim 314 misto 3,14 ziskame vysledek 100krat vétsi.
Abychom dostali celou ¢ast vysledku musime S délit 100 (fadek 3). Zbytek, ktery vypocitame po
odecteni celociselné casti nasobené 100, pak musime zobrazit i s pfipadnou prvni nulou (fadek 6).

Priklad 3:

Zadani:
Vytvorte program, ktery pod sebe vypiSe vSechna licha ¢isla do 20.

Rada:
Pouzijte prikazy PRINT, LET, GOTO a IF-THEN

Priklad reSeni:

0 REM Licha cisla do 20
1LETA=1

2 PRINTA, “\n“

3LET A=A+2

4 IF A<20 THEN GOTO 2

Poznamky k reSeni:

Jde o velmi jednoduchy pocitany cyklus. Zaciname od 1 (fadek 1) a postupné cisla vypisujeme na
samostatné fadky (fadek 2) se zvétSovanim po 2 (fadek 3). Vyhodnocovaci podminka (fadek 4)
vraci program na vypis ¢isla, dokud neni dosaZena 20.

Priklad 4:

Zadani:
Vytvorte program, ktery vykresli na obrazovce pravouhly trojihelnik ze znakt * vysoky 10 radku.

Rada:
PouZijte prikazy PUTCH, LET, GOTO a IF-THEN. Znak pro novy fadek ma hodnotu 10 a znak *
ma hodnotu 42.

Priklad reSeni:

0 REM Trojuhelnik z *

1 PUTCH 10

2 LET A=1; REM Pocita radky

3 LET B=1; REM Hvezdy na radku
4 PUTCH 42; LET B=B+1

5IF B<=A THEN GOTO 4

6 PUTCH 10; LET A=A+1

7 IF A<11 THEN GOTO 3

Poznamky k reSeni:

Jsou zde pouZity 2 pocitané cykly vnotrené do sebe. V proménné A se pocitaji fadky (fadky 2,6,7) a
v proménné B se pocitaji znaky vypisované na radku (fadky 3,4,5).

Priklad 5:

Zadani:
Vytvorte program, ktery vykresli na obrazovce ¢tverec o délce strany 10 znakd.

Rada:

PouZijte prikazy PUTCH, PRINT, LET, GOTO a IF-THEN. Abyste si uSetfili praci, pouZijte pro
vykresleni vodorovné ¢ary podprogram s prikazy GOSUB a RETURN. Nezapomeiite program ve
spravném bodé ukoncit prikazem END.

Znak pro vymazani displeje méa hodnotu 12, novy fadek 10, levy horni roh 144, pravy horni roh
145, levy dolni roh 146, pravy dolni roh 147, svisla ¢ara 156 a vodorovna Cara 157.

Priklad reSeni:

0 REM Ctverec

1 PUTCH 12

2 LET A=1; REM Pocita radky

3 PUTCH 144; GOSUB 10

4 PUTCH 145; PUTCH 10

5 PUTCH 156; PRINT “ “

6 PUTCH 156; PUTCH 10; LET A=A+1
7 IF A<8 THEN GOTO 5

8 PUTCH 146; GOSUB 10

9 PUTCH 147; PUTCH 10; END
10 REM Vodorovna cara

11 LET I=1

12 PUTCH 157; LET I=I+1

13 IF I<9 THEN GOTO 12

14 RETURN

Poznamky k reSeni:

Radky 3 a 4 vykresli horni stranu ¢tverce, fadky 8 a 9 spodek &tverce. Vyuziva se podprogram na
fadcich 10 aZ 14, ktery déld vodorovnou ¢aru. Radky 5, 6 a 7 vykresli svislé ¢ary. Pocet mezer v
prikazu PRINT na fadku 5 musi byt 8, aby se vyprazdnilo télo ¢tverce.

Priklad 6:

Zadani:
Vytvorite program, ktery umozni posunovat znak X po obrazovce. Zacina se uprostied obrazovky a
stisknuti klavesy 2 znak X posune dol{, 4 vlevo, 6 vpravo a 8 nahoru.

Rada:

PouZzijte prikaz PUTCH s vyuZitim jeho parametri, CURSOR, INKEY, GOTO a IF-THEN.

Znak pro vymazani displeje ma hodnotu 12, znak X je 88, prazdny znak je 32, 2 je 50, 4 je 52, 6 je
54 a 8 je 56.

Priklad reSeni:

0 REM Posun X

1 PUTCH 12

2 CURSOR 400; PUTCH .88

3 INKEY K

4 IF K=50 THEN PUTCH v32; PUTCH .88
5 IF K=52 THEN PUTCH <32; PUTCH .88
6 IF K=54 THEN PUTCH >32; PUTCH .88
7 IF K=56 THEN PUTCH A32; PUTCH .88
8 GOTO 3

Poznamky k feSeni:

Program nejprve vycisti obrazovku a umisti doprostred displeje (pozice 400) znak X pomoci
piikazu PUTCH. Symbol tecky zajisti, Ze kurzor ztistane na dané pozici. KdyzZ uZivatel stiskne
spravnou klavesu, nejprve se vymaZze stara pozice prazdnym znakem a posune se kurzor na novou
pozici (smér urcuje symbol pred hodnotou 32). Na nové pozici se pak vypiSe X a ponecha kurzor.

Priklad 7:

Zadani:

Vytvorite program, ktery bude automaticky ,,hrat ping-pong“. Micek v podobé pismene o bude
poletovat na radku z levé strany na pravou stranu a naopak. Na kraji displeje se vZdy odrazi a vrati
Zpét.

Rada:

Pouzijte pfikaz PUTCH s vyuZitim parametrid, CURSOR, DELAY, GOTO a IF-THEN.

Znak pro vymazani displeje ma hodnotu 12, znak o je 111, prazdny znak je 32.

Pozice uprostfed prostredniho radku ma hodnotu 400, znak na levém okraji 384, vpravo 415.

Priklad reSeni:

0 REM Ping-pong

1 PUTCH 12; LET C=400

2 CURSOR C; PUTCH .111

3LET D=1

4 DELAY 100; PUTCH .32

5 LET C=C+D; CURSOR C; PUTCH .111
6 IF C=415 THEN LET D=-1

7 IF C=384 THEN LET D=1

8 GOTO 4

Poznamky k feSeni:

Nejprve se program nastavi do vychozi pozice a smér pohybu micku v proménné D se urci na
hodnotu 1 — tj. micek poleti vpravo(fadky 1 aZ 3). Ve smycce, ktera za¢ina na radku 4, se chvili
pocka, smaZe stary obraz micku a pak se posune micek podle hodnoty v D na novou pozici. Radky
6 a 7 hlidaji dosaZeni kraje obrazovky a méni smér letu tam (D=1), nebo zpét (D=-1).

Priklad 8:

Zadani:
Napiste program, ktery zahraje melodii podle not na obrazku. Na zacatku uZivatel zada dobu trvani
1/8 noty v rozmezi 200 aZz 400 milisekund.

PribliZzné frekvence zakladnich tonu:
c: 262 d‘: 294 e‘: 330 fe: 349 g 392 a‘: 440 h¢: 494
c“: 523 d“: 587 e“: 659 f: 698 g“: 784 a“: 880 h*: 989

Rada:

PouZzijte prikaz BEEP s vyuzitim parametrd, INPUT, GOTO a IF-THEN. Pokud se v melodii
objevuji stejné pasaze, miZete pouzit podprogram GOSUB a RETURN, abyste nemuseli opisovat
stejné Casti. Nezapomefite program ukoncit pfikazem END.

Priklad reSeni:

0 REM Melodie

1 PRINT “Zadej tempo:\n“

2 INPUTT

3 IF T<200 THEN GOTO 1

4 IF T>400 THEN GOTO 1

5 GOSUB 15

6 BEEP 523,2*T; BEEP 494,4*T; GOSUB 15

7 BEEP 587,2*T; BEEP 523,4*T; BEEP 392, T

8 BEEP 392, T; BEEP 784,2*T; BEEP 659,2*T

9 BEEP 523,2*T; BEEP 494,2*T; BEEP 440,2*T
10 BEEP 698,T; BEEP 698,T; BEEP 659,2*T

11 BEEP 523,2*T; BEEP 587,2*T; BEEP 523,4*T
12 END

15 BEEP 392,T; BEEP 392,T; BEEP 440,2*T

16 BEEP 392,2*T; RETURN

Poznamky k feSeni:

UZivatel je nejprve vyzvan k zadani tempa a je zkontrolovano povolené rozmezi hodnot (fadky 1 az
4). ProtoZe se prvni Ctyfi tony opakuji ve tfetim taktu, je pouZit podprogram na radcich 15 a 16.
Jednotlivé tény jsou zadany frekvenci a dobou trvani. Ctvrt'ové nota je dvojnasobna a ptilova
Ctyfnasobna proti zadané dobé osminové noty.

Priklad 9:

Zadani:

NapiSte program pro vytvoreni velmi jednoduchého klaviru. Po stisknuti klavesy 1 az 9 zahraje po
dobu 200 aZ 600 milisekund (Cas se zada po startu) ton c azZ d“.

Priblizné frekvence zakladnich tonti:

c‘: 262 d: 294 e‘: 330 f¢: 349 g‘: 392 a‘: 440 h¢: 494
c“: 523 d“: 587
Rada:

Pouzijte pfikaz BEEP s vyuzitim parametrti, INPUT, INKEY, GOTO a IF-THEN. Zkuste vyuZit
toho, Ze klavesy 1 az 9 vraci v prikazu INKEY hodnoty 49 az 57. Pak miiZete vypocitat potfebny
skok v ptfikazu GOTO.

Priklad feSeni:

0 REM Piano

1 PRINT “Zadej dobu tonu:\n“
2INPUTT

3 IF T<200 THEN GOTO 1
4 IF T>600 THEN GOTO 1
5 INKEY K

6 IF K<49 THEN GOTO 5
7 IF K>57 THEN GOTO 5
8 GOTO K-40

9 BEEP 262,T; GOTO 5

10 BEEP 294,T; GOTO 5
11 BEEP 330,T; GOTO 5
12 BEEP 349,T; GOTO 5
13 BEEP 392,T; GOTO 5
14 BEEP 440,T; GOTO 5
15 BEEP 494,T; GOTO 5
16 BEEP 523,T; GOTO 5
17 BEEP 587,T; GOTO 5

Poznamky k feSeni:

UZivatel je nejprve vyzvan k zadani doby tonu a je zkontrolovano povolené rozmezi hodnot (fadky
1 aZ 4). Ve smycce, ktera za¢ina na fadku 5 se nejprve zkontroluje, zda klavesa patfi mezi povolené
a pak se vypocita na radku 8 skok na radek s odpovidajicim tonem. Hodnota prvni povolené cislice
49 musi zamiFit na fadek 9, proto se odecita hodnota 40.

Priklad 10:

Zadani:
Vytvorte program, ktery bude predstavovat losovaci zatizeni pro €isla 1 azZ 49. VZdy po stisknuti
libovolné klavesy zobrazi nahodné cislo z uvedeného rozsahu.

Rada:

Pouzijte prikazy TIME, INKEY, PRINT, GOTO a IF-THEN.

Po stisknuti klavesy vezmeéte aktualni cas. Jeho zbytek po déleni 49 bude dostatecné nahodné cislo.
Nezapomeiite, Ze zbytek po déleni 49 je v rozsahu 0 aZ 48. Ziskanou hodnotu je potfeba zvysit o 1.

Priklad reSeni:

0 REM Nahodne cislo

1 INKEY K

2 IF K=0 THEN GOTO 1
3 TIME T; LET A=T/49
4 PRINT T-A*49+1, “\n*
5GOTO 1

Poznamky k feSeni:

Na radcich 1 a 2 se ceka na stisknuti libovolné klavesy. V okamziku, kdy uZivatel stiskne klavesu se
naplni proménnd T prikazem TIME. Hodnota ziskana ptikazem TIME se méni ptibliZné 32000krat
za sekundu. ProtoZe potfebujeme nahodna Cisla do 49, vypocitame zbytek po déleni 49. Vysledek
zvétSeny o 1 zobrazime.

Rev. B



Priklad 11:

Zadani:
Vytvorte program ukazujici ¢as uplynuly od spusténi pocitace ve formé HH:MM:SS.

Rada:

Pouzijte prikazy TIME, PRINT, PUTCH, CURSOR, GOTO a IF-THEN.

Pokud hodnotu ziskanou prikazem TIME podélite 31914, ziskate cas od spusSténi pocitace v
sekundach. Ziskany Cas pak staci délit 3600, abyste dostali pocet uplynulych hodin. Ze zbytku pak
délenim 60 ziskate minuty. Nezapomeiite, Ze v zadani je zobrazeni ¢isla na 2 mista. Nestaci vypsat
jenom napf. 5, je tfeba zobrazit jako 05. Na to se mtiZe hodit podprogram s GOSUB a RETURN.
Znak : ma hodnotu 58, znak 0 ma hodnotu 48.

Priklad reSeni:

0 REM Hodiny

1 PUTCH 12

2 DELAY 300

3TIME T; LET S=T/31914

4 LET A=S5/3600; LET S=S-A*3600
5 CURSOR 0; GOSUB 10

6 LET A=S5/60; LET S=S-A*60

7 PUTCH 58; GOSUB 10

8 LET A=S; PUTCH 58; GOSUB 10
9 GOTO 2

10 REM Vypis z A na 2 mista

11 IF A<10 THEN PUTCH 48

12 PRINT A; RETURN

Poznamky k reSeni:

Cas ziskany piikazem TIME podélime hodnotou 31914, abychom ziskali celkovy ¢as v sekundach.
Hodiny do proménné A ziskame délenim 3600. Ve zbytku po celoc¢iselném déleni zlistanou minuty a
sekundy. Celoc¢iselnym délenim 60 dostaneme minuty a zbytek z vypoctu minut jsou sekundy. V3e
musime zobrazovat na 2 mista, takZe dopliiujeme pfipadnou prvni nulu u jednocifernych cisel pri
zobrazovani hodnoty z A v podprogramu (fadek 11).

Priklad 12:

Zadani:

Vytvoite pomoci pocitace ,,kuchyiiskou minutku®. Po startu programu uZivatel zada ¢as od 1 do 99
minut. Po zadani zacne na displeji odpocitavani po minutach a po dosaZeni 0 se ozve 10 sekund
signal.

Rada:

Pouzijte ptikaz TIME, BEEP, INPUT, PRINT, GOTO a IF-THEN. Po zadani poZadovaného Casu si
miiZete vypocitat Cas, pri kterém casovani skonci. Cilovy cas ziskate prictenim pocCtu minut
nasobenych 60*31914 (sekundy v minuté krat pocet tikd za sekundu = 1914840) k aktudlnimu casu.

Priklad reSeni:

0 REM Minutka

1 PRINT “Zadej cas [min]:\n“

2 INPUTT

3IF T<1 THEN GOTO 1

4 IF T>99 THEN GOTO 1

5 TIME C; LET C=C+T*60*31914

6 PUTCH 12

7 DELAY 300

8 TIME T; LET D=C-T

9 IF D<0 THEN BEEP 800,10000; END
10 CURSOR 0; PRINT D/1914840, « «
11 GOTO 7

Poznamky k reSeni:

UZivatel je nejprve vyzvan k zadani doby a je zkontrolovano povolené rozmezi hodnot (fadky 1 az
4). Po spravném zadani je do proménné C uloZen cilovy Cas, ktery vypocitame jako aktualni cas s
dobou zadanou uzivatelem vynasobenou 31914 (pocet tiki za sekundu) a 60 (pocet sekund v
minuté). Ve smycce, ktera zacina na radku 7, chvili cekame, protoZe neni nutno zobrazovat hodnoty
tak rychle, Ze by to uZivatel stejné nepostehl. Na fadku 8 zjistime rozdil mezi cilovym a aktualnim
Casem. Pri zaporném vysledku casovani ukonc¢ime. Jinak rozdil zobrazime. Konstanta 1914840
vznikne vynasobenim 60*31914.

Priklad 13:

Zadani:

Naprogramujte hru ,,L.epsi postfeh® pro dva a vice hraci. Po stisknuti klavesy ¢eka 5 sekund,
pricem? start Casovani je zahajen *. Pak zobrazi na displeji cilovy znak. Klavesa, ktera byla
zmacknuta jako prvni po zobrazeni cilového znaku, bude oznacena jako vitéz.

Rada:

Pouzijte ptikazy INKEY, DELAY, PUTCH, PRINT, GOTO a IF-THEN. Pf¥i tvorbé programu
nezapomeiite na to, Ze musite zkontrolovat, zda hra¢ nezmacknul klavesu pred zobrazenim cilového
znaku.

Znak * ma hodnotu 42, cilovy znak ma hodnotu 143.

Priklad reSeni:

0 REM Lepsi postreh

1 PRINT “\nStart hry\n*

2 INKEY K

3 IF K=0 THEN GOTO 2

4 PUTCH 42; DELAY 5000; INKEY K

5 PUTCH 143; IF K>0 THEN GOTO 10

6 INKEY K

7 IF K=0 THEN GOTO 6

8 PRINT “\nVitez: “; PUTCH K; DELAY 2000
9GOTO 1

10 PRINT “\nProhral: “; PUTCH K; DELAY 2000
11 GOTO 1

Poznamky k feSeni:

Na radcich 1 az 3 pockame na start hry. Pak zobrazime *, cekame 5 sekund a sejmeme, zda nékdo
nestiskl klavesu predcasné (fadek 4). Na fadku 5 zobrazime cilovy znak a zkontrolujeme predcasné
stisknuti klavesy. Je-li vSe v poradku, na radcich 6 a 7 ¢ekame na stisknuti klavesy a na radku 8
vyhlasime vitéze.

Priklad 14:

Zadani:
Vytvorte program pro akusticky generator. UZivatel zada frekvenci poZadovaného ténu v rozsahu
100 az 4000 Hz a dokud nebude program ukoncen, pocitac piska zadanym toénem.

Rada:

Pouzijte ptikazy BEEP, INPUT, PRINT, GOTO a IF-THEN.

Aby bylo moZné program zastavit zadanim prikazu END, pouZijte v asovém parametru u prikazu
BEEP néjakou pfijatelnou hodnotu v rozsahu jednotek sekund.

Priklad reSeni:

0 REM Zvukovy generator

1 PRINT “Zadej frekv. 100-4000\n*
2 INPUTF

3 IF F<100 THEN GOTO 1

4 TF F>4000 THEN GOTO 1

5 BEEP F,5000

6 GOTO 5

Poznamky k feSeni:

UZivatele nejprve vyzveme k zadani frekvence a je zkontrolovano povolené rozmezi hodnot (fadky
1 aZ 4). Pak je ve smycce na fadcich 5 a 6 trvale poustén tén o zadané frekvenci.

Priklad 15:

Zadani:
NapiSte program, ve kterém uZivatel zada néjaké jméno ukoncené klavesou Enter. Pocitac pak
vypiSe, Ze jméno je jeho pritel. Maximalni délka jména miZe byt 30 znaki.

Rada:
PouZijte prikazy INKEY, PRINT, PUTCH, GOTO a IF-THEN. Zadavané jméno je vyhodné ukladat
do proménnych v poli @. Pocitejte si pritom pocet uloZenych pismen.

Priklad reSeni:

0 REM Pritel pocitace

1 LET I=0; PRINT “\nZadej jmeno:\n“
2 INKEY K

3 IF K=0 THEN GOTO 2

4 TF K=10 THEN GOTO 10

5 PUTCH K; LET @I=K; LET I=I+1
6 IF [=30 THEN GOTO 10

7 GOTO 2

10 LET J=0; PUTCH 10

11 PUTCH @J; LET J=J+1

12 IF J<ITHEN GOTO 11

13 PRINT “ je muj pritel\n“

Poznamky k reSeni:

Radky 2 a 3 ¢ekaiji na zadani znaku. Radek 4 kontroluje, zda nebylo zaddvéani ukonceno klavesou
Enter. Na fadku 5 je znak uloZen do pole. Radek 6 kontroluje maximalni povolenou délku jména.
Na Fadcich 10 aZ 12 je jméno opét vypsano.

Priklad 16:

Zadani:
Vytvoite program pro ovladani LED pfipojené k vystupu D1 pomoci kladvesnice. Prvni zmacknuti
libovolné klavesy LED rozsviti, druhé zhasne, dalsi opét rozsviti atd.

Doplnkovy HW:
Na vystup D1 pfipojte LED s rezistorem podle nasledujiciho obrazku. Cervena barva LED bude
vhodna pro nasledujici ulohu.

D1

478R

'C4

RED

GND

Rada:

Pouzijte prikazy INKEY, DOUT, GOTO a IF-THEN. Prikaz DOUT 1,1 pak LED rozsviti a DOUT
0,1 ji zhasne.

ProtoZe program nemusi byt rychly, miiZete do cyklu cekani na stisk klavesy vlozit drobné zpoZdéni
piikazem DELAY.

Priklad reSeni:

0 REM Ovladani LED
1LETL=0

2 DELAY 100; INKEY K

3 IF K=0 THEN GOTO 2

4 TF L=0 THEN GOTO 6

5 LET L=0; DOUT 0,1; GOTO 2
6 LET L=1; DOUT 1,1; GOTO 2

Poznamky k reSeni:

Proménna L uklada aktualni hodnotu vystupu, abychom védéli, jak pristé vystup zmeénit. Na fadcich
2 a 3 se Ceka na stisk klavesy. Radek 4 rozhodne podle stavu L, zda se LED rozsviti nebo zhasne.

Priklad 17:

Zadani:
Vytvorte program, ktery z TinyBasRV udéla ovladani vystraznych svétel na Zelezni¢nim prejezdu i
se zvukovou signalizaci.

Doplnkovy HW:
Na vystupy D1 a D2 pripojte dvé LED (nejlépe Cervené) s rezistory podle nasledujiciho obrazku.

D1

02

478R 470r

W
W

GND

Rada:
PouZijte prikazy BEEP, DELAY, DOUT a GOTO. Zvukovou signalizaci si vyzkouSejte a upravte
podle vlastnich predstav.

Priklad reSeni:

0 REM Zeleznice - vystraha

1 DOUT 1,1; DOUT 0,2

2 BEEP 2000,100; BEEP 800,200; DELAY 200
3DOUT 0,1; DOUT 1,2

4 BEEP 2000,100; BEEP 800,200; DELAY 200
5GOTO 1

Poznamky k reSeni:

Radky 1 a 3 rozsviti vzdy pouze jednu z obou LED. Na fadcich 2 a 4 je generovano stejné zvukové
znameni. Uvedeny priklad jen velmi vzdalené pfipomind cinkani zvonce. MizZete je ladit a
upravovat podle svého ucha...

Priklad 18:

Zadani:
Vytvorite program pro méreni vstupniho napéti na vstupu AIN v mV. Na vstup pfived’te napéti z
béZce potenciometru, které bude zobrazovano s periodou %2 sekundy.

Doplnkovy HW:
Na vystup AIN pripojte linedrni potenciometr s hodnotou odporu v rozmezi 1 az 100 kQ podle
nasledujiciho obrazku.

+5V

AIN

10K

GND

Rada:

PouZijte prikazy AINP, DELAY, CURSOR, PRINT a GOTO. Hodnota ziskana z A/D prevodniku je
v rozsahu 0 aZ 1023. To predstavuje napéti 0 aZ 5V — tedy 0 aZ 5000 mV. Abychom prepocitali
rozsah A/D prevodniku na mV, musime hodnotu nejprve vynasobit 5000 a pak délit 1023.

Priklad FeSeni:

0 REM Analog vstup mV

1 PUTCH 12

2 CURSOR 397; AINP A

3 PRINT A*5000/1023, “mV
4 DELAY 500

5 GOTO 2

Poznamky k feSeni:

Na fadku 2 se nejprve nastavi kurzor priblizné do stfedu obrazovky a pak se zméfi napéti na vstupu
AIN. Pfepocet se provede pri vypisu hodnoty v prikaze PRINT podle navodu v odstavci Rada. Po
vypisu jednotek jsou jeSté doplnény 3 mezery, aby se vymazal pfipadny delSi zapis z minulého
méfeni.

Rev. B



Priklad 19:

Zadani:
Upravte pocitac¢ TinyBasRV tak, aby pracoval jako jednoduchy voltmetr s rozsahem do 50 V. Napéti
bude zobrazovano s periodou %2 sekundy a s rozliSenim na 2 desetinna mista.

Doplnkovy HW:
Na vystup AIN pripojte vstupni odporovy délic s rezistory s pomérem odport 9:1. Idealni by bylo
napt. 900 kQ a 100 kQ. Hodnota 900 kQ je ale mimo béZnou fadu dostupnych rezistorti. MiiZete

pouZit nejblizsi hodnotu 910 kQ z fady E24. V nouzi lze pouZit i 1 MQ a upravit konstantu v
programu.

+ INPUT

916K

-

O GND
- INPUT

Rada:

Pouzijte ptikazy AINP, DELAY, CURSOR, PUTCH, PRINT, GOTO a IF-THEN. Hodnota ziskana
AINP z A/D prevodniku je v rozsahu 0 az 1023. To predstavuje napéti 0 aZ 50V (musime zobrazit
0.00 aZ 50.00 V). Abychom prepocitali rozsah A/D prevodniku na 'V, je tfeba hodnotu nejprve
vynasobit 5000 a pak délit 1023. Vysledné Cislo zobrazime s desetinou teckou mezi cislicemi.
Nasobici konstanta 5000 je idealni pripad, kdy se pfedpoklada presné referenc¢ni napéti v
mikropocitaci a presné hodnoty odporu obou rezistorti v délici. V praxi se to bude liSit. Proto je
potieba u konkrétniho kusu pocitace tuto hodnotu upravit tak, aby zobrazené napéti se co nejvice
bliZilo napéti zmérenému referencnim voltmetrem.

Priklad reSeni:

0 REM Voltmetr 50V

1 PUTCH 12; LET K=5320
2 CURSOR 397; AINP A

3 LET V=A*K/1023; LET C=V/100
4 LET Z=V-C*100

5PRINT C, “.

6 IF Z<10 THEN PRINT “0*
7 PRINT Z

8 DELAY 500

9 GOTO 2

Poznamky k reSeni:

Na fadku 1 je nastavena prepoctova konstanta, kterou si musite upravit podle konkrétniho kusu
pocitace a vstupniho odporového délice. DalSi postup jiZ kopiruje instrukce uvedené v odstavci
Rada. Nejprve je vypsana cela ¢ast z proménné C a pak zbytek po déleni 100 ze Z. Pokud je zbytek
mensi neZ 10, je nutné doplnit uvodni nulu (fadek 6).

Priklad 20:

Zadani:
Pomoci expanderu PCF8575 pripojeného k TinyBasRV vyrobte béZici svétlo na fadé LED
pripojenych k vystuptim expanderu.

Doplnkovy HW:
Na vystupy modulu s expanderem PSC8575 pripojte vice LED podle obrazku. ZaleZi na mnozstvi
LED, které mate k dispozici, maximaln€é je moZné pFipojit 16 LED.

+5V

Rada:

PouzZijte prikazy I2CW, DELAY, LET, GOTO a IF-THEN. Pokud chcete rozsvitit prvni LED,
musite do expanderu poslat binarni ¢islo b1111111111111110 nebo dekadicky -2. Druha LED
vyZaduje poslat binarni ¢islo b1111111111111101 nebo dekadicky -3. Dalsi by bylo binarni ¢islo
b1111111111111011 nebo dekadicky -5. Rozsviceni LED se posune na dalSi pozici vZdy
vynasobenim 2 a prictenim 1.

Po rozsviceni posledni LED se musi svétlo vratit na zacatek.

Priklad feSeni (pro 8 LED):

0 REM Bezicich 8 LED

1LET A=-2

2 12CW A; DELAY 200

3 LET A=A*2+1

4 1F A <-300 THEN LET A=-2
5GOTO 2

Poznamky k FeSeni:

Program postupuje podle instrukci uvedenych v odstavci Rada. Po dany priklad bylo pouZito pouze
8 LED, pri dosaZeni hodnoty mensi neZ -300 se hodnota v A vrati na zacatek. Poslani ¢isla -257 do
expanderu (posledni odeslané v cyklu) na chvili vypne vSech 8 LED. Pokud byste to nechtéli,
zmenSete podminku napf. na A<-150.

Priklad 21:

Zadani:
Napiste program, ktery pomoci expanderu PCF8575 pripojeného k TinyBasRV rozsviti po stisknuti
klavesy 0 az 9 urceny pocet LED (vami zvoleny obraz).

Doplnkovy HW:
Na vystupy modulu s expanderem PSC8575 pripojte vice LED podle obrazku. ZaleZi na mnozstvi
LED, které mate k dispozici, maximaln€é je moZné pFipojit 16 LED.

+5V

Rada:

PouZijte piikazy I2CW, DELAY, INKEY, GOTO a IF-THEN. Pro jednotlivé Cislice si miiZete urcit
libovolny vzorek rozsvicenych LED. Napf. po stisknuti klavesy 1 se rozsviti kazda licha LED
prikazem [2CW b1010101010101010, klavesa 2 rozsviti dolni polovinu LED prikazem [2CW
b1111111100000000 atd.

Klavesy 0 az 9 vraci v prikazu INKEY hodnoty 48 az 57.

Priklad reSeni:

0 REM Vzory z LED

1 DELAY 100; INKEY K

2 IF K<48 THEN GOTO 1

3 IF K>57 THEN GOTO 1

4 GOTO K-43

512CW b1100110011001100; GOTO 1
6 12CW b1010101010101010; GOTO 1
7 12CW b1111111100000000; GOTO 1
8 I2CW b1111100001100101; GOTO 1
9 I2CW b0000000011001100; GOTO 1
10 I2CW b1100110000000000; GOTO 1
11 I2CW b1110111011101110; GOTO 1
12 I2CW b1101110111011101; GOTO 1
1312CW b1011101110111011; GOTO 1
14 12CW b0111011101110111; GOTO 1

Poznamky k FeSeni:

Nejprve se na fadcich 1, 2 a 3 zajisti pfijeti pouze povolené &iselné klavesy. Radek 4 pak zajisti
odskok na patficny vzorek rozsvicenych LED s navratem na zacatek.

Priklad 22:

Zadani:

Naprogramujte méri¢ postifehu uZivatele. Po stisknuti klavesy ¢eka program nahodny pocet sekund
(1 aZ 10), pricem? start Casovani je zahajen *. Pak zobrazi na displeji cilovy znak. UZivatel musi co
nejrychleji zmacknout libovolnou klavesu. Na displeji se zobrazi reakéni ¢as v milisekundach.

Rada:

Pouzijte prikazy INKEY, TIME, DELAY, PUTCH, PRINT, GOTO a IF-THEN. Pfi tvorbé programu
nezapomeiite na to, Ze musite zkontrolovat, zda hra¢ nezmacknul klavesu pred zobrazenim cilového
znaku. Pro méfeni ¢asu v milisekundach budeme pocitat, Ze TIME vraci kaZzdou milisekundu
hodnotu o 32 vétsi.

Znak * ma hodnotu 42, cilovy znak ma hodnotu 143.

Priklad reSeni:

0 REM Meric postrehu

1 PRINT “\nStart\n“

2 INKEY K

3 IF K=0 THEN GOTO 2

4 TIME T; LET C=T/10

5 LET D=T-C*10+1; LET D=D*1000
6 PUTCH 42; DELAY D

7 INKEY K; PUTCH 143; TIME S
8 IF K>0 THEN GOTO 14

9 INKEY K

10 IF K=0 THEN GOTO 9

11 TIME E; LET R=E-S

12 PRINT “\nReakce: “, R/32

13 PRINT “ milisec*; GOTO 1

14 PRINT “\nPredcasne!“

15 GOTO 1

Poznamky k reSeni:

Radky 2 a 3 ¢ekaji na start méfeni. Pak se ziska pomoci pfikazu TIME a zbytku po déleni 10
nahodny cas, ktery se uloZi do proménné D v milisekundach (fadky 4 a 5). Na fadku 6 se zobrazi *
a provede ¢ekani. Radek 7 sejme stav klaves, zobrazi cilovy znak a uloZi zacate¢ni ¢as pro méfeni
postfehu. Na fadku 8 se zkontroluje pfipadné predcasné stisknuti klavesy. Radky 9 a 10 Cekaji na
reakci uZivatele. Cas, kdy zareagoval se uloZi do E a pak na Fadcich 11, 12 a 13 se vysledek vypise.

Priklad 23:
Zadani:
Mate-li moZnost pripojit k TinyBasRV jehlickovou tiskarnu s paralelnim portem pres expander

PCF8575, vytvorte z ni jednoduchy psaci stroj.

Doplnkovy HW:
Na vystupy modulu s expanderem PCF8575 pripojte konektor Cannon 25 samici podle obrazku.

GND 15

Rada:

PouZijte prikazy I2CW, DELAY, INKEY, GOTO a BEEP. Pfikaz BEEP se pouZije pro zvukovou
signalizaci stisknuti klavesy kratkym pipnutim. Neni nutné jej pouZit.

Na vyvodu 1 paralelniho portu je signal STROBE ovladany vystupem P10, ktery potvrdi a zapiSe
data do tiskarny. Aktivni hodnota je logicka 0. Po zapsani dat, je nutné tento signal vratit do klidové
hodnoty 1. To Ize nejsnaze udélat zapsanim hodnoty o 256 vétsi neZ zapisovana data.

Priklad reSeni:

0 REM Psaci stroj

1 INKEY K

2 IF K=0 THEN GOTO 1
312CW K; PUTCH K

4 DELAY 10

5 [2CW K+256; BEEP 1000,50
6 GOTO 1

Poznamky k feSeni:

Radky 1 a 2 ekaji na stisk klavesy. Radek 3 posle znak do tiskarny a okopiruje ho na displej. Na
fadku 4 se chvili pocka (pro pripad pomalé reakce tiskarny). Na fadku 5 se musi nastavit signal
STROBE pro tiskarnu zpét na logickou 1 (hodnota znaku se zvysi o 9ty bit nastaveny na 1) a kratce
se pipne.

Rev. B



Priklad 24:

Zadani:

Vytvoite z TinyBasRV jednoduchou celociselnou kalkulacku. UZivatel zada prvni ¢islo, pak
operand (+ - * /) a na zavér druhé cislo. Na displeji se objevi vypocitany vysledek ve tvaru napr.
256*2=512.

Rada:

Pouzijte piikazy INPUT, INKEY, PRINT, PUTCH, GOTO a IF-THEN. Znaky operandt maji
nasledujici hodnoty:

+ 43 - 45 * 42 / 47
Nezapomeiite, Ze nulou nedélime!

Priklad reSeni:

0 REM Kalkulacka

1 PUTCH 10; INPUT A

2 INKEY K

3 IF K=42 THEN GOTO 8

4 IF K=43 THEN GOTO 8

5IF K=45 THEN GOTO 8

6 IF K=47 THEN GOTO 8

7 GOTO 2

8 PUTCH K; PUTCH 10; INPUT B;

9 PRINT A; PUTCH K; PRINT B

10 IF K=43 THEN PRINT “=“ A+B; GOTO 1
11 IF K=45 THEN PRINT “=“,A-B; GOTO 1
12 IF K=42 THEN PRINT “=“ A*B; GOTO 1
13 IF B=0 THEN PRINT “ Nulou nelze!*“; GOTO 1
14 PRINT “=“,A/B; GOTO 1

Poznamky k reSeni:

Prvni hodnota se uloZi do proménné A na fadku 1. Pak se na fadcich 3 aZ 7 ¢eka na typ operace. Na
fadku 8 vytiskneme znak operace a pockdme na druhou hodnotu. Radky 10 aZ 14 provedou vypocet
podle stavu K, pricemz se na fadku 13 zkontroluje, zda se nebude délit 0.

Priklad 25:

Zadani:

NapiSte program pro hadani tajného cisla 0 az 99. Po startu si pocita¢ vybere ndhodné cislo a hrac
ma toto ¢islo uhodnout na co nejmensi pocet pokusii. Po kazdém pokusu pocitac uvede, zda je tajné
Cislo vétsi nebo mensi nez to, které zadal hrac.

Rada:
PouZijte prikazy INPUT, PRINT, TIME, LET, GOTO a IF-THEN. Nejprve si vygenerujte pomoci
ptikazu TIME a zbytku po déleni 100 nahodné cislo.

Priklad reSeni:

0 REM Hadani cisla

1 TIME T; LET C=T/100; LET X=T-C*100
2 LET P=0; PRINT “\nHadej cislo\n“

3 INPUT A; LET P=P+1

4 TF A=X THEN GOTO 8

5IF A>X THEN GOTO 7

6 PRINT P,“. Hadej vic\n“; GOTO 3

7 PRINT P,“. Hadej mene\n“; GOTO 3

8 PRINT “Uhodl jsi na “, P, “ pokusu\n“

Poznamky k feSeni:

Na fadku 1 se pomoci prikazu TIME a zbytku po celociselném déleni 100 uloZi do proménné X
nahodné generované hadané ¢islo. Na fadcich 2 a 3 je uZivatel vyzvan k pokusu uhodnout ¢islo a v
proménné P se pocitaji provedené pokusy. Radky 4 a 5 kontroluji ispésnost pokusu a na zbyvajicich
fadcich se vypisuje pokyn, kam sméfovat dalSi odhad nebo ukonceni programu s vysledkem.

Priklad 26:

Zadani:
Vytvorte z TinyBasRV jednoduchy kédovy zamek. Zada-li uZivatel na klavesnici spravnou
posloupnost 4 nebo 5 ¢islic, na vystupu D1 se na 3 sekundy objevi logicka 1 a zazni zvukovy signal.

Rada:

Pouzijte prikazy INKEY, LET, DOUT, BEEP, GOTO a IF-THEN. Pocet kédovych cislic si miiZete
zvolit sami. Tajnou ¢iselnou kombinaci si uloZte do pole @ a kaZda spravné uhodnuta cislice
posune ukazatel v poli na vyssi hodnotu. Pti chybé se ukazatel vrati na pocatek pole.

Klavesy 0 az 9 vraci v pfikazu INKEY hodnoty 48 azZ 57.

Priklad reSeni:

0 REM Kodovy zamek

1 LET @0=49,57,50,48,51
2LET P=0

3 INKEY K

4 IF K=0 THEN GOTO 3

5 IF K<>@P THEN GOTO 2
6 LET P=P+1;

7 IF P<5 THEN GOTO 3

8 DOUT 1,1; BEEP 800,3000; DOUT 0,1
9 GOTO 2

Poznamky k feSeni:

Na radku 1 je do pole @ uloZen kod zamku. Proménna P ukazuje, kam se uZivatel pri zadavani
kédu dostal. Radek 5 rozhoduje, zda se postoupi na dalsi pozici, nebo se vrati opét na prvni &islici
kodu. Na radku 7 se pri spravném postupu a dosaZeni ukazatele P na hodnotu 5 umozni prejit na
fadek 8, kde se kodovy zamek na 3 sekundy otevre.

Priklad 27:

Zadani:
NapiSte program, ktery bude nahrazovat hraci kostku. Program bude na displeji zobrazovat body
tak, jak se nachazeji na kostce.

Rada:

Pokud se podivéte na vzory nachéazejici se na sténach kostky, zjistite, Ze kromé stfedni tecky se
teCky vZzdy vyskytuji po dvojicich. Napf. 3 je kombinace A+B, 5 je A+B+C.

=) () (e
o

Vzory si miZete zakddovat do pole napr. pomoci binarnich Cisel a pak je podle ndhodné taZeného
Cisla zobrazit. Doporucené pozice kurzoru jsou pro A: 400, pro B: 235 a 565, pro C: 245 a 555 a pro
D: 395 a 405. Pouzijte prikazy LET, TIME, INKEY, GOTO, CURSOR, PUTCH a IF-THEN.

eB

B

Priklad reSeni:

0 REM Hraci kostka

1 PUTCH 12; LET @0=1,2,3,6,7,14

2 INKEY K

3 IF K=0 THEN GOTO 2

4 TIME D; LET C=D/6; LET D=D-C*6

5 CURSOR 400; PUTCH .32; IF @D&1 THEN PUTCH 143
6 CURSOR 235; PUTCH 32; CURSOR 565

7 PUTCH .32; IF@D&?2 THEN PUTCH 143

8 CURSOR 235; IF@D&2 THEN PUTCH 143
9 CURSOR 245; PUTCH 32; CURSOR 555

10 PUTCH .32; IF@D&4 THEN PUTCH 143
11 CURSOR 245; IF@D&4 THEN PUTCH 143
12 CURSOR 395; PUTCH 32; CURSOR 405
13 PUTCH .32; IF@D&8 THEN PUTCH 143
14 CURSOR 395; IF@D&8 THEN PUTCH 143
15 GOTO 2

Poznamky k reSeni:

V poli @ je zakddovano zobrazeni Cisel 0 az 5 pomoci binarniho kédu. Pokud se podivate, jak se
uvedena Cisla zapisuji ve dvojkové soustavé, zjistite, Ze nejniZsi bit predstavuje teCku A, vyssi bit
obé tecky B, dalsi C a ¢tvrty bit jsou tecky D. Radky 2 a 3 ¢ekaji na stisknuti klavesy. Na fadku 4 je
vygenerovano do proménné D nahodné Cislo 0 aZ 5. Pak jsou prochazeny vSechny pozice
jednotlivych teCek. Napf. u fadku 5 je kurzor nastaven na teCku A a ta je nejprve smazana. Pokud
prvek pole s indexem D ma nastaveny nejniZsi bit, je teCka na pozici vykreslena.

Rev. B



Priklad 28:

Zadani:

Dopliite TinyBasRV tak, aby fungoval jako Fidici panel nouzového osvétleni. Pomoci fotorezistoru
se bude snimat stav okolniho osvétleni a méfena hodnota se zobrazi na displeji. UZivatel zada na
zacatku hodnoty pri kterych se zapne a naopak vypne LED pfipojena na D1.

Doplikovy HW:
Na vstup AIN a vystup D1 pripojte d€lic s fotorezistorem a LED s rezistorem podle obrazku.

+5V

'Cs

AIN

D1

33K

GND!

Rada:

Pouzijte ptikazy INPUT, PRINT, DOUT, AINP, GOTO a IF-THEN. Méfena hodnota osvétleni bude
pouze Cislo z A/D prevodniku. UzZivatel si miiZe nastavit 2 hranice, pfi kterych se svétlo zapne nebo
vypne, a tak si nastavit hysterezi obvodu. OSettete, aby zadané hranice nebyly mimo rozumny
rozsah.

Jako fotorezistor miizete pouZit typ 5537 nebo podobny.

Priklad reSeni:

0 REM Nouzove osvetleni

1 PRINT “\MfO\n\n“

2 INPUT H

3 IF H<10 THEN GOTO 2

4 IF H>1000 THEN GOTO 2
5 INPUT L

6 IF L>=H THEN GOTO 5

7 AINP A; CURSOR 0; PRINT A,«“
8 IF A<L THEN DOUT 1,1
9 IF A>H THEN DOUT 0,1
10 DELAY 200; GOTO 7

Poznamky k FeSeni:

Na fadcich 2 az 6 je vyZadovano zadani hranic, pfi kterych LED zhasne nebo se rozsviti (i s
kontrolou povolenych hodnot). Vlastni smycka je na fadcich 7 az 10. Po zméreni je hodnota
vypsana. Pfikaz PRINT je doplnén o vymazani delSiho cisla, které by mohlo ztistat na displeji z
predchoziho méteni. Pak je na fadcich 8 a 9 LED rozsvicena ¢i zhasnuta.

Priklad 29:

Zadani:

Vytvorte z TinyBasRV jednoduchy kryptograficky pristroj, ktery bude kédovat a dekédovat znaky
abecedy a aZ z na jiny znak podle vami zvolené Sifrovaci tabulky. Program po stisku prislusného
znaku musi ukazat jak zakddovany, tak odpovidajici dekddovany znak.

Rada:

V poli @ si pomoci prikazu LET vytvorte Sifrovaci tabulku. Po kontrole, zda zmacknuta klavesa
patii mezi povolené znaky, vyberte znak z tabulky, zobrazte ho a pak tabulku prohledejte pro
dekoddovani. Doporucené prikazy PRINT, INKEY, CURSOR, GOTO a IF-THEN.

Priklad reSeni:

0 REM Sifrovaci stroj

1LET @0=6,18,11,5,20,14,22,2,9,15

2 LET @10=17,8,1,10,19,0,13,4,23,24
3LET @20=25,7,16,12,21,3

4 PRINT “\fZnak Koder Dekoder

5 INKEY K

6 IF K<97 THEN GOTO 5

7 IF K>122 THEN GOTO 5

8 LET Z=K-97; CURSOR 34; PUTCH K
9 CURSOR 41; PUTCH 65+@Z; LET I=0
10 IF @I=7Z THEN GOTO 12

11 LET I=I+1; IF I<26 THEN GOTO 10
12 CURSOR 50; PUTCH 65+I; GOTO 5

Poznamky k feSeni:

Zékladem programu je kédovaci tabulka v poli @. VSech 26 znakt anglické abecedy ma sviij
nahodné zvoleny kod. Znak ,,a“ mé kéd na pozici 0, ,,b“ na 1, atd. Cisla kédt mohou byt od 0 do
25.

Uvazujeme zadavani malych pismen abecedy, jejich hodnoty budou 97 az 122. Na to ¢ekaji radky 5
az 7. Ukazatel do kodovaci tabulky v proménné Z ziskame odectenim hodnoty 97 od znaku.
VypiSeme zadany znak, jeho zakédovanou podobu ve formé velkého pismene, které zacinaji na
hodnotéach 65 (fadky 8 a 9). Na radcich 10 a 11 tabulku prohledame pro opacnou funkci —
dekédovani. Radek 12 dekédovany znak vypise.

Priklad 30:

Zadani:
NapiSte program pro jednoduché stopky. Prvni zmacknuti libovolné klavesy méreny cas odstartuje,
dalsi pak hodnotu zastavi. Zobrazovat se bude na minuty:sekundy,desetiny sekund.

Rada:

Pouzijte ptikazy INKEY, LET, PUTCH, PRINT, GOTO a IF-THEN. Po spusténi programu nejprve
pockejte na zmacknuti klavesy. UloZte si aktualni Cas startu. Zobrazujte rozdil mezi aktualnim
Casem a startem déleni 3191 (pocet tikii za desetinu sekundy). Nejprve hodnotu délenou 600 — to
budou minuty a pak sekundy a desetiny sekund. Nezapomeiite, zobrazit sekundy vZdy na 2
desetinna mista. Znak 0 ma hodnotu 48.

Dal3i stisk pocitani zastavi, aby se dal precist dosazeny cCas. Treti stisk klavesy pak opét vrati stopky
na nulovy cas a pripravi ke startu.

Priklad reSeni:

0 REM Stopky

1 PUTCH 12; PRINT “0:00,0¢

2 INKEY K

3 IF K=0 THEN GOTO 2

4 TIME B

5 DELAY 80

6 TIME T; LET T=T-B; LET S=T/3191
7 LET A=S/600; LET S=S-A*600

8 CURSOR 0; PRINT A,“:“

9 LET A=S/10; LET S=S-A*10

10 IF A<10 THEN PUTCH 48

11 PRINT A,“,“,S

12 INKEY K; IF K=0 THEN GOTO 5
13 INKEY K

14 IF K=0 THEN GOTO 13

15 GOTO 1

Poznamky k reSeni:

Reseni postupuje podle pokyni v odstavci Rada. Nejprve se ¢eka na start stopek (fadky 2 a 3). Do
proménné B se uloZi Cas startu. Lidské oko neni prilis rychlé, tak ma smycka zobrazovani zanesené
zpoZdéni na fadku 5. Uplynuly Cas v desetindch sekundy se vypocita do proménné S na fadku 6.
Radky 7 aZ 11 ¢as zobrazi na displej. Radek 12 kontroluje zastaveni stopek. Radky 13 a 14 ¢ekaji
na novy restart programu.

Priklad 31:

Zadani:
Napiste program predstavujici jednoduchou hru ,,Dostihy“. Hra¢ mtize s pocatecnim kapitalem
vsadit castku na jednoho ze tfi koni. Koné budou do cile dobihat s nahodnym poradim.

Rada:

Pouzijte napf. prikazy INPUT, PRINT, PUTCH, LET, CURSOR, DELAY, TIME, GOTO a IF-
THEN. Pozici, na které se nachazi kiin, si uloZte do pole @. Bude se pak 1épe odkazovat na koné,
ktery se bude ndhodné posouvat pomoci pfikazu TIME.

P¥i sazce na vitézného koné pripocitejte hraci trojnasobek vsazené castky.

Priklad reSeni:

0 REM Dostihy

1 LET M=1000

2 PRINT “\fVsadis? 1-“,M,“\n“

3 INPUT B; IF B<1 THEN GOTO 2

4 IF B>M THEN GOTO 2

5 PRINT “\nKun 1-3?%; INPUT K

6 LET @1=1; LET @2=1; LET @3=1

7 PRINT “\fZavod“; PUTCH p49,64

8 PUTCH p50,128; PUTCH p51,192

9TIME T; LET V=T/3; LET V=T-V*3+1

10 DELAY 100; LET @V=@V+1; PUTCH p143,V*64+@V
11 IF @V<31 THEN GOTO 9

12 CURSOR 395; PRINT “Vyhral “,V; DELAY 1000
13 IF V=K THEN LET M=M+B*3; GOTO 2

14 LET M=M-B; IF M>0 THEN GOTO 2

15 PRINT “\nDosly prachy\n“

Poznamky k feSeni:

Pocatecni kapital je nastaven na fadku 1. Pak je uZivatel vyzvan, aby urcil hodnotu sazky s
kontrolou, zda se nepokousi $vindlovat (fadky 2 aZ 4). Radek 5 vyzve k sazce na jednoho ze tfi
koritli. Zde se platnost volby nekontroluje, pokud hrac zada nesmyslné ¢islo, penize vyhodi oknem.
Radek 6 usadi koné na startovni pozici (pozice koni jsou v odpovidajicim prvku pole @). Radky 7 a
8 pripravi grafiku zavodu. Na fadku 9 se vygeneruje, ktery kil ndhodné postoupi o jedno policko.
To je vykresleno na fadku 10 s tim, Ze pozice koné se uloZi do jeho proménné.

Radek 11 Fesi pripadné vitézstvi koné a ukonceni zdvodu. Na fadcich 12 aZ 15 se vyplaci vyhra,
odebiraji prohrané penize a pripadné ukoncuje hra pfi nedostatku penéz.

Priklad 32:

Zadani:

Napiste pro TinyBasRV jednoduchou hru ,,Zabi skokani®. Na zac¢atku budou 4 Zaby jednoho druhu
vlevo a 4 Zaby druhého druhu vpravo. Uprostted je jedna mezera. Hra¢ miiZe skocit s libovolnou
Zabou do vedlejsi mezery nebo jednu sousedni Zabu do mezery preskocit. Cilem je na co nejmensi
pocet skokti premistit Zaby zleva na pravou stranu a naopak.

Rada:

Pouzijte ptikazy INPUT, LET, PUTCH, PRINT, CURSOR, GOTO a IF-THEN. Pocatecni a cilovou
pozici si miZete uloZit do pole @. Znaky pro Zaby vlevo a vpravo si zvolte sami. Hra¢ bude muset
vZdy zadat ,,odkud” skace , kam®. JednodusSe to Ize zadavat ¢islem napt. 34 — skok z pozice 3 na 4.
Budete muset oSetfit, aby hrac nezadal nesmyslné pozice skoku. Nesmi také skakat pres 2 a vice Zab
naraz.

Priklad reSeni:

0 REM Zabi skokani

1LET P=0; LET @1=8,8,8,8,15,1,1,1,1

2 LET @10-1,1,1,1,15,8,8,8,8

3 PUTCH 12; LET I=1

4 CURSOR I*2; PUTCH v80+@]I; PRINT I

5 LET I=I+1; IF I<10 THEN GOTO 4

6 PRINT “\n\n“; INPUT M; IF M>99 THEN GOTO 3
7 LET F=M/10; LET T=M-F*10; LET D=F-T
8 IF T<1 THEN GOTO 3

9 IF F<1 THEN GOTO 3

10 IF D<0 THEN LET D=-D

11 IF @T<>15 THEN GOTO 3

12 LET I=1; IF D>2 THEN GOTO 3

13 LET P=P+1; LET @T=@F; LET @F=15
14 LET K=1+9; IF @I<>@K THEN GOTO 3
15 LET I=I+1; IF I<10 THEN GOTO 14

16 PRINT “\nUspech na “,P,“ skoku*

Poznamky k reSeni:

V proménné P je uloZen pocet skoki. V poli jsou nejprve uloZeny pozice Zab pfi startu a pak i
cilové pozice. Cisla jsou mala aby se snadno v programu zapsala. Pro zobrazeni budou znaky
zvySeny o 80, takZe levé Zaby se objevi jako Q (81), pravé jako X (88) a volna pozice bude
podtrzitko (95). Na fadcich 3 aZ 5 je vykresleno hraci pole s Zabami i s jejich pozicemi.

Na fadku 6 hrac zada sviij skok. Thned se kontroluje, zda ¢islo neni moc velké. Na fadku 7 se
rozdéli odkud (F) a kam (T) se skace a jak dlouhy je to skok (D). Radky 8 a 9 kontroluji, zda nebyla
nékde zadana 0 nebo zaporné ¢islo. Radek 10 opravi délku skoku na kladné ¢islo.

Radek 11 kontroluje, jestli se skace do volné pozice, fadek 12, neni-li skok moc dlouhy. Na fadku
13 se provede platny skok. Radky 14 a 15 zkontroluji pfipadny konec hry pfi dosaZeni cile.

